

```

end;
UHEN;
for I:=0 step 1 until M do
  begin K:=H×F[I]; QI:=Q[I];
    R:=(K-2.0×QI)/6.0;
    Q[I]:=QI+3.0×R-0.5×K;
    Y[I]:=Y[I]+R
  end;
UHEN;
INSATU
end

```

〔注 1〕 H には h を入れる。K には $h f_i$ を、R には r_{ij} を、QI には q_{ij} を (一時的に)、それぞれ入れる。N でステップ数を数え、NN にはその最終値を入れておく。I には i 、M には m を入れる。Y[I] には y_{ij} 、Q[I] には q_{ij} 、F[I] には $f_i(y_0, y_1, \dots, y_m)$ を入れる。

〔注 2〕 UHEN (右辺) という名の手続は $y_0, y_1, \dots, y_m$ に対する f_i の値を計算するものである。ここでは具体例として $f_i(y_0, y_1, y_2) = y_2$ 、 $f_2(y_0, y_1, y_2) = y_0 y_2 + y_1$ の場合が示してある。INSATU (印刷) という名の手続は、まず x を第 1 行に、それから y_i, f_i ($i = 1, \dots, m$) をならべて m 行にわたって印刷するものである。

〔注 3〕 入力データとしては、キザミ h 、全ステップ数 NN、従属変数の数 m 、初期値 $a, \eta_1, \dots, \eta_m$ を、この順序に与える。

〔注 4〕 プログラム中の定数の由来

$$\begin{aligned} 0.29289322 &= 1 - \sqrt{1/2}, \\ 1.70710678 &= 1 + \sqrt{1/2}. \end{aligned}$$

6302. 連立常微分方程式の解法 (Milne 法)

森口繁一 (東大工学部)

微分方程式

$$\frac{dy_i}{dx} = f_i(x, y_1, \dots, y_m) \quad (i=1, 2, \dots, m)$$

を、初期条件

$$y_i(a) = \eta_i \quad (i=1, \dots, m)$$

の下で解くものとする。

x を h キザミで増し、 $x_n = a + nh$ とする。 $x = x_n$ に対する y_i の値を $y_{i,n}$ 、そして、そこでの右辺の値を $f_{i,n}$ と書く。予測の公式は

$$y_{i,n+4}^p = y_{i,n} + \frac{4}{3} h (2f_{i,n+1} - f_{i,n+2} + 2f_{i,n+3}),$$

修正の公式は

$$y_{i,n+3}^c = y_{i,n+1} + \frac{h}{3} (f_{i,n+1} + 4f_{i,n+2} + f_{i,n+3})$$

である (ただし、予測のあと番号 n を 1 だけ進めてから修正の公式を使う)。第 1 回の修正量 C_{0i} は精度の目安になるので記録しておく。修正量が、あらかじめ指定した定数 ϵ よりも小さくなったら収束したものととして次に進む。

プログラム

MILNERENRITU:

```

begin real H, EPS, X, YP, YC, C;
integer N, NN, I, M, NC, S1, S2;
array Y0, Y1, Y2, F1, F2, Y, F, C0[1:10];
procedure UHEN;
  begin F[1]:=Y[2];
    F[2]:=X×Y[2]+Y[1]
  end;
procedure INSATU;
  begin CRLF; PRINTSTRING('X=');
    PRINTREAL(X);
    PRINTSTRING('NC=');
    PRINTINTEGER(NC);
    for I:=1 step 1 until M do
      begin CRLF; PRINTREAL(Y[I]);
        PRINTREAL(F[I]); PRINTREAL(C0[I])
      end
    end;

```

procedure SYUPPATUTI;

```

begin for I:=1 step 1 until M do
  READREAL(Y[I]);
  UHEN; INSATU
end;

```

```

HAZIME: READREAL(EPS); READREAL(H);
CRLF; PRINTREAL(EPS);
PRINTREAL(H); CRLF;
READINTEGER(NN); READINTEGER(M);
NC:=0; for I:=1 step 1 until M do
  C0[L]:=0.0;
  READREAL(X); SYUPPATUTI;
  for I:=1 step 1 until M do Y0[I]:=Y[I];
  X:=X+H; SYUPPATUTI;
  for I:=1 step 1 until M do
    begin Y1[I]:=Y[I]; F1[I]:=F[I] end;
  X:=X+H; SYUPPATUTI;
  for I:=1 step 1 until M do
    begin Y2[I]:=Y[I]; F2[I]:=F[I] end;
  X:=X+H; SYUPPATUTI;

```

SINKO: for N:=4 step 1 until NN do

```

begin X:=X+H; NC:=0; S1:=0;

```

YOSOKU: for I:=1 step 1 until M do

```

begin YP:=Y0[I]+(2.0×F1[I]-F2[I]
  +2.0×F[I])×H/0.75;

```

ZENSIN: Y0[I]:=Y1[I]; Y1[I]:=Y2[I];

```

      Y2[I] := Y[I];
      Y[I] := YP; F1[I] := F2[I]; F2[I] := F[I]
    end;
SYUSEI: UHEN; NC := NC + 1; S2 := 0;
    for I := 1 step 1 until M do
      begin YC := Y1[I] + (F1[I] + 4.0 × F2[I]
        + F[I]) × H/3.0;
        C := YC - Y[I]; Y[I] := YC;
        if S1 = 0 then C0[I] := C;
        if ABS(C) ≥ EPS then S2 := 1
        end;
        S1 := 1;
        if S2 ≠ 0 then go to SYUSEI;
        UHEN; INSATU
      end
    end
end

```

〔注 1〕 H に h , EPS に ε , X に x を入れる。YP には予測値 $y_{i,n+1}^P$ を, YC には修正値 $y_{i,n+1}^C$ を, C には修正量を, いずれも

一時的に, それぞれ入れる。N でステップ数を数え, NN にはその最終値を入れておく。I には i を, M には m を入れる。NC で修正の回数を数える。S1 は最初の修正のときだけ 0 で, その他のときは 1 となる。S2 は収束の判定に用い, すべての i に対して修正量が ε 以下になっているかどうかを $S2=0$ かどうかで見分ける。

Y0[I], Y1[I], Y2[I] にはそれぞれ $y_{i,n}$, $y_{i,n+1}$, $y_{i,n+2}$ を入れ, F1[I], F2[I] にはそれぞれ $f_{i,n+1}$ および $f_{i,n+2}$ を入れる。Y[I] と F[I] には一般に y_i と f_i とを入れるが, 予測および修正のときはとくに $y_{i,n+1}$ と $f_{i,n+1}$ が入っている (一般に UHEN という手続では X, Y[1], ..., Y[M] から計算した右辺の値を F[1], ..., F[M] へ記憶するようにしてある)。C0[I] には初回の修正量 C_{0i} を入れる。

〔注 2〕 UHEN は右辺 f_i を計算する手続。ここには一例として, $m=2$ で, $f_1(x, y_1, y_2) = y_2$, $f_2(x, y_1, y_2) = x y_2 + y_1$ の場合が示してある。INSATU は印刷の手続で, まず第 1 行に x と, その右に修正の回数を印刷し, それから y_i, f_i, C_{0i} をならべて m 行に印刷する。SYUPPATUTI という名の手続は, 出発値 $y_{i,n}, i=1, \dots, m$ を読み込み, 右辺を計算し, 印刷を行なう。これは $n=0, 1, 2, 3$ のそれぞれについて呼び出される。

〔注 3〕 入力データとしては, 精度 ε , キザミ h , 全ステップ数 NN, 従属変数の数 m , 始点 a , 出発値 $(y_{1,n}, \dots, y_{m,n}) (n=0, 1, 2, 3)$ を, この順序に与える。出発値のうち, $y_{1,0}, \dots, y_{m,0}$ は初期値 $\eta_1, \dots, \eta_m$ に一致するが, その他は別の方法 (たとえば Runge-Kutta-Gill 法) で求めたものを使う。

(昭和 38 年 4 月 18 日受付)